

# SWI-Prolog **cplint** Pack Manual

Fabrizio Riguzzi  
fabrizio.riguzzi@unife.it

January 10, 2016

## 1 Introduction

**cplint** is a suite of programs for reasoning with LPADs/CP-logic programs [10], [11], [8], [9]. It contains modules for both inference and learning.

## 2 Installation

**cplint** is distributed as a pack of SWI-Prolog. To install it, use

```
?- pack_install(cplint).
```

Moreover, in order to make sure you have a foreign library that matches your architecture, run

```
?- pack_rebuild(cplint).
```

## 3 Syntax

LPAD and CP-logic programs consist of a set of annotated disjunctive clauses. Disjunction in the head is represented with a semicolon and atoms in the head are separated from probabilities by a colon. For the rest, the usual syntax of Prolog is used. A general CP-logic clause has the form

```
h1:p1 ; ... ; hn:pn :- b1,...,bm,\+ c1,...,\+ cl
```

No parentheses are necessary. The **pi** are numeric expressions. It is up to the user to ensure that the numeric expressions are legal, i.e. that they sum up to less than one.

If the clause has an empty body, it can be represented like this

```
h1:p1 ; ... ; hn:pn.
```

If the clause has a single head with probability 1, the annotation can be omitted and the clause takes the form of a normal prolog clause, i.e.

```
h1 :- b1,...,bm,\+ c1,...,\+ cl.
```

stands for

```
h1:1 :- b1,...,bm,\+ c1,...,\+ cl.
```

The coin example of [11] is represented as (file `coin.cpl`)

```
heads(Coin):1/2 ; tails(Coin):1/2 :-  
    toss(Coin),\+biased(Coin).
```

```
heads(Coin):0.6 ; tails(Coin):0.4 :-  
    toss(Coin),biased(Coin).
```

```
fair(Coin):0.9 ; biased(Coin):0.1.
```

```
toss(coin).
```

The first clause states that if we toss a coin that is not biased it has equal probability of landing heads and tails. The second states that if the coin is biased it has a slightly higher probability of landing heads. The third states that the coin is fair with probability 0.9 and biased with probability 0.1 and the last clause states that we toss a coin with certainty.

Moreover, the bodies of rules may contain the built-in predicates:

```
is/2, >/2, </2, >=/2, <=/2,  
:=/2, =\=/2, true/0, false/0,  
=/2, ==/2, \=/2, \==/2, length/2
```

The bodies may also contain the following library predicates:

```
member/2, max_list/2, min_list/2  
nth0/3, nth/3, dif/2, select/3
```

plus the predicate

```
average/2
```

that, given a list of numbers, computes its arithmetic mean.

## 4 Inference

`cplint` answers queries using the module `pita`. It performs the program transformation technique of [7]. Differently from that work, techniques alternative to tabling and answer subsumption are used.

For answering queries, you have to prepare a Prolog file where you first load `pita` and then enclose the probabilistic clauses in `:-cplint.` and `:-end_cplint.` For example, the coin program above can be stored in `coin.pl` as follows

```
:- use_module(library(pita)).  
:- cplint.
```

```
heads(Coin):1/2 ; tails(Coin):1/2:-  
toss(Coin),\+biased(Coin).
```

```
heads(Coin):0.6 ; tails(Coin):0.4:-  
toss(Coin),biased(Coin).
```

```
fair(Coin):0.9 ; biased(Coin):0.1.
```

```
toss(coin).
```

```
:- end_cplint.
```

Then you can simply load `coin.pl` as

```
?- [coin].
```

and query it with

```
?- prob(heads(coin),P).
```

Note that supplying `coin.pl` as an argument to the `swipl` command currently returns errors due to bad interaction between `pita.pl` and the top-level. The program is loaded correctly anyway but it is recommended to load it from the top-level to avoid these errors.

## 4.1 Parameters

The module makes use of a number of parameters in order to control its behavior. They can be set with the directive

```
:- set_pita(<parameter>,<value>).
```

inside the couple `:-cplint.` and `:-end_cplint.`

The current value can be read with

```
?- setting_pita(<parameter>,Value).
```

from the top-level. The available parameters are:

- **epsilon\_parsing**: if (1 - the sum of the probabilities of all the head atoms) is larger than **epsilon\_parsing**, then **pita** adds the null event to the head. Default value 0.00001.
- **single\_var**: determines how non ground clauses are treated: if **true**, a single random variable is assigned to the whole non ground clause, if **false**, a different random variable is assigned to every grounding of the clause. Default value **false**.
- **depth\_bound**: if **true**, the depth of the derivation of the goal is limited to the value of the **depth** parameter. Default value **false**.
- **depth**: maximum depth of derivations when **depth\_bound** is set to **true**. Default value 2.

## 4.2 Parameters

The module makes use of a number of parameters in order to control its behavior. They can be set with the directive

```
:- set_pita(<parameter>,<value>).
```

inside the couple `:-cplint.` and `:-end_cplint.`

The current value can be read with

`?- setting_pita(<parameter>,Value).`

from the top-level. The available parameters are:

- **epsilon\_parsing**: if (1 - the sum of the probabilities of all the head atoms) is larger than **epsilon\_parsing**, then **pita** adds the null event to the head. Default value 0.00001.
- **single\_var**: determines how non ground clauses are treated: if **true**, a single random variable is assigned to the whole non ground clause, if **false**, a different random variable is assigned to every grounding of the clause. Default value **false**.
- **depth\_bound**: if **true**, the depth of the derivation of the goal is limited to the value of the **depth** parameter. Default value **false**.
- **depth**: maximum depth of derivations when **depth\_bound** is set to **true**. Default value 2.

## 5 Learning

The following learning algorithms are available:

- **EMBLEM** (EM over Bdds for probabilistic Logic programs Efficient Mining): an implementation of EM for learning parameters that computes expectations directly on BDDs [3], [1], [2]
- **SLIPCOVER** (Structure LearnIng of Probabilistic logic programs by searching OVER the clause space): an algorithm for learning the structure of programs by searching the clause space and the theory space separately [4]

### 5.1 Input

To execute the learning algorithms, prepare a Prolog file divided in five parts

- preamble
- background knowledge, i.e., knowledge valid for all interpretations
- LPAD/CPL-program for you which you want to learn the parameters (optional)
- language bias information
- example interpretations

The preamble must come first, the order of the other parts can be changed.

For example, consider the Bongard problems of [6]. The `pack/cplint/prolog/examples/learning` folder in SWI-Prolog home contains some example learning files. For example, it contains `bongard.pl` and `bongardkeys.pl` that represent a Bongard problem. Let us consider `bongard.pl`.

### 5.1.1 Preamble

In the preamble, the SLIPCOVER library is loaded with

```
:- use_module(library(slipcover)).
```

Then, if you are using your file in cplint on SWISH, you could add

```
:- if(current_predicate(use_rendering/1)).  
:- use_rendering(c3).  
:- use_rendering(lpad).  
:- endif.
```

if you want a nice representation of the output (in particular, if you want graphs of the ROC and PR curves).

Now you can initialize SLIPCOVER with

```
:- sc.
```

At this point you can start setting parameters for SLIPCOVER such as for example

```
:- set_sc(megaex_bottom,20).  
:- set_sc(max_iter,2).  
:- set_sc(max_iter_structure,5).  
:- set_sc(verbosity,1).
```

We will see later the list of available parameters. A particularly important parameter is `verbosity`: if set to 1, nothing is printed and learning is fastest, if set to 3 much information is printed and learning is slowest, 2 is in between. This ends the preamble.

### 5.1.2 Background and Initial LPAD/CPL-program

Now you can specify the background knowledge with a fact of the form

```
bg(<list of terms representing clauses>).
```

and an initial program with a fact of the form

```
in(<list of terms representing clauses>).
```

The initial program is used in parameter learning for providing the structure. The indicated parameters do not matter as they are first randomized. Remember to enclose each clause in parentheses because `:-` has the highest precedence.

For example, `bongard.pl` has the initial program

```
in([(pos:0.197575 :-  
    circle(A),  
    in(B,A)),  
    (pos:0.000303421 :-  
    circle(A),  
    triangle(B)),  
    (pos:0.000448807 :-  
    triangle(A),  
    circle(B))]).
```

Both facts should be present. If there are no background/input clauses then write `bg([])./in([]).`

### 5.1.3 Language Bias

The language bias part contains the declarations of the input and output predicates. Output predicates are declared as

```
output(<predicate>/<arity>).
```

and indicate the predicate whose atoms you want to predict. Derivations for the atoms for this predicates in the input data are built by the systems.

Input predicates are those whose atoms you are not interested in predicting. You can declare closed world input predicates with

```
input_cw(<predicate>/<arity>).
```

For these predicates, the only true atoms are those in the interpretations, the clauses in the input/hypothesized program are not used to derive atoms not present in the interpretations.

Open world input predicates are declared with

```
input(<predicate>/<arity>).
```

In this case, if a subgoal for such a predicate is encountered when deriving a subgoal for the output predicates, both the facts in the interpretations and the clauses of the input program are used.

Then, you have to specify the language bias by means of mode declarations in the style of Progol.

```
modeh(<recall>,<predicate>(<arg1>,...)).
```

specifies the atoms that can appear in the head of clauses, while

```
modeb(<recall>,<predicate>(<arg1>,...)).
```

specifies the atoms that can appear in the body of clauses. **<recall>** can be an integer or \*. **<recall>** indicates how many atoms for the predicate specification are retained in the bottom clause during a saturation step. \* stands for all those that are found. Otherwise the indicated number is randomly chosen.

Arguments of the form

```
+<type>
```

specifies that the argument should be an input variable of type **<type>**, i.e., a variable replacing a **+<type>** argument in the head or a **-<type>** argument in a preceding literal in the current hypothesized clause.

Another argument form is

```
-<type>
```

for specifying that the argument should be a output variable of type **<type>**. Any variable can replace this argument, either input or output. The only constraint on output variables is that those in the head of the current hypothesized clause must appear as output variables in an atom of the body.

Other forms are

```
#<type>
```

for specifying an argument which should be replaced by a constant of type `<type>` in the bottom clause but should not be used for replacing input variables of the following literals when building the bottom clause or

`-#<type>`

for specifying an argument which should be replaced by a constant of type `<type>` in the bottom clause and that should be used for replacing input variables of the following literals when building the bottom clause.

`<constant>`

for specifying a constant.

An example of language bias for the Bongard domain is

`output(pos/0).`

```
input_cw(triangle/1).
input_cw(square/1).
input_cw(circle/1).
input_cw(in/2).
input_cw(config/2).
```

```
modeh(*,pos).
modeb(*,triangle(-obj)).
modeb(*,square(-obj)).
modeb(*,circle(-obj)).
modeb(*,in(+obj,-obj)).
modeb(*,in(-obj,+obj)).
modeb(*,config(+obj,-#dir)).
```

SLIPCOVER also requires facts for the `determination/2` predicate Aleph-style that indicate which predicates can appear in the body of clauses. For example

```
determination(pos/0,triangle/1).
determination(pos/0,square/1).
determination(pos/0,circle/1).
determination(pos/0,in/2).
determination(pos/0,config/2).
```

state that `triangle/1` can appear in the body of clauses for `pos/0`.

SLIPCOVER also allows mode declarations of the form

```
modeh(<r>,[<s1>,...,<sn>],[<a1>,...,<an>],[<P1/Ar1>,...,<Pk/Ark>]).
```

These mode declarations are used to generate clauses with more than two head atoms. In them, `<s1>,...,<sn>` are schemas, `<a1>,...,<an>` are atoms such that `<ai>` is obtained from `<si>` by replacing placeholders with variables, `<Pi/Ar1>` are the predicates admitted in the body. `<a1>,...,<an>` are used to indicate which variables should be shared by the atoms in the head. An example of such a mode declaration (from `uwcselearn.pl`) is

```
modeh(*,
[advisedby(+person,+person),tempadvisedby(+person,+person)],
[advisedby(A,B),tempadvisedby(A,B)],
[professor/1,student/1,hasposition/2,inphase/2,
publication/2,taughtby/3,ta/3,courselevel/2,yearsprogram/2]).
```

#### 5.1.4 Example Interpretations

The last part of the file contains the data. You can specify data with two modalities: models and keys. In the models type, you specify an example model (or interpretation or megaexample) as a list of Prolog facts initiated by `begin(model(<name>)).` and terminated by `end(model(<name>)).` as in

```
begin(model(2)).
pos.
triangle(o5).
config(o5,up).
square(o4).
in(o4,o5).
circle(o3).
triangle(o2).
config(o2,up).
in(o2,o3).
triangle(o1).
config(o1,up).
end(model(2)).
```

The interpretations may contain a fact of the form

```
prob(0.3).
```

assigning a probability (0.3 in this case) to the interpretations. If this is omitted, the probability of each interpretation is considered equal to  $1/n$  where  $n$  is the total number of interpretations. `prob/1` can be used to set a different multiplicity for the interpretations.

The facts in the interpretation are loaded in SWI-Prolog database by adding an extra initial argument equal to the name of the model.

Alternatively, with the keys modality, you can directly write the facts and the first argument will be interpreted as a model identifier. The above interpretation in the keys modality is

```
pos(2).
triangle(2,o5).
config(2,o5,up).
square(2,o4).
in(2,o4,o5).
circle(2,o3).
triangle(2,o2).
config(2,o2,up).
in(2,o2,o3).
triangle(2,o1).
config(2,o1,up).
```

which is contained in the `bongardkeys.pl` This is also how model 2 above is stored in SWI-Prolog database. The two modalities, models and keys, can be mixed in the same file.

Note that you can add background knowledge that is not probabilistic directly to the file writing the clauses taking into account the model argument. For example `carc.pl` contains



```
connected(_M, Ring1, Ring2):-
    Ring1 \= Ring2,
    member(A, Ring1),
    member(A, Ring2), !.
```

```
symbond(Mod, A, B, T):- bond(Mod, A, B, T).
symbond(Mod, A, B, T):- bond(Mod, B, A, T).
```

where the first argument of all the atoms is the model.

Then you must indicate how the examples are divided in folds with facts of the form: `fold(<fold_name>, <list of model identifiers>)`, as for example

```
fold(train, [2, 3, ...]).
fold(test, [490, 491, ...]).
```

## 5.2 Commands

### 5.2.1 Parameter Learning

To execute EMBLEM, prepare an input file as indicated above (for example `bongard.pl`), load it into SWI-Prolog

```
?- [bongard].
```

and call

```
?- induce_par([train], P).
```

to obtain in `P` the program in input with the updated parameters.

The results can also be tested on a test set with

```
?- induce_par([train], [test], P, LL, AUCROC, ROC, AUCPR, PR).
```

This returns the log likelihood of the test examples in `LL`, the Area Under the ROC curve in `AUCROC`, a dictionary containing the list of points (in the form of Prolog pairs `x-y`) of the ROC curve in `ROC`, the Area Under the PR curve in `AUCPR`, a dictionary containing the list of points of the PR curve in `PR`.

The general form of the commands is

```
?- induce_par(<list_of_train_folds>, P).
```

```
?- induce_par(<list_of_train_folds>, <list_of_test_folds>,
P, LL, AUCROC, ROC, AUCPR, PR).
```

### 5.2.2 Structure Learning

To execute SLIPCOVER, prepare an input file as indicated above (for example `bongard.pl`), load it into SWI-Prolog

```
?- [bongard].
```

and call

```
?- induce([train], P).
```

to obtain in `P` the learned program.

The results can also be tested on a test set with

```
?- induce([train],[test],P,LL,AUCROC,ROC,AUCPR,PR).
```

returning the log likelihood of the test examples in `LL`, the Area Under the ROC curve in `AUCROC`, a dictionary containing the list of points (in the form of Prolog pairs `x-y`) of the ROC curve in `ROC`, the Area Under the PR curve in `AUCPR`, a dictionary containing the list of points of the PR curve in `PR`.

### 5.3 Parameters

Parameters are set with commands of the form

```
:- set_sc(<parameter>,<value>).
```

The available parameters are:

- **depth\_bound**: (values: `{true,false}`, default value: `true`) if `true`, the depth of the derivation of the goal is limited to the value of the **depth** parameter.
- **depth** (values: integer, default value: 2): depth of derivations if **depth\_bound** is set to `true`
- **single\_var** (values: `{true,false}`, default value: `false`): if set to `true`, there is a random variable for each clause, instead of a different random variable for each grounding of each clause
- **epsilon\_em** (values: real, default value: 0.1): if the difference in the log likelihood in two successive parameter EM iteration is smaller than **epsilon\_em**, then EM stops
- **epsilon\_em\_fraction** (values: real, default value: 0.01): if the difference in the log likelihood in two successive parameter EM iteration is smaller than **epsilon\_em\_fraction**\*(-current log likelihood), then EM stops
- **iter** (values: integer, default value: 1): maximum number of iteration of EM parameter learning. If set to -1, no maximum number of iterations is imposed
- **iterREF** (values: integer, default value: 1, valid for SLIPCOVER): maximum number of iteration of EM parameter learning for refinements. If set to -1, no maximum number of iterations is imposed.
- **random\_restarts\_number** (values: integer, default value: 1, valid for EMBLEM and SLIPCOVER): number of random restarts of parameter EM learning
- **random\_restarts\_REFnumber** (values: integer, default value: 1, valid for SLIPCOVER): number of random restarts of parameter EM learning for refinements
- **setrand** (values: `rand(integer,integer,integer)`): seed for the random functions, see SWI-Prolog manual for the allowed values

- **logzero** (values: negative real, default value  $\log(0.000001)$ ): value assigned to  $\log 0$
- **max\_iter** (values: integer, default value: 10, valid for SLIPCOVER): number of iterations of beam search
- **max\_var** (values: integer, default value: 4, valid for SLIPCOVER): maximum number of distinct variables in a clause
- **beamsize** (values: integer, default value: 100, valid for SLIPCOVER): size of the beam
- **megaex\_bottom** (values: integer, default value: 1, valid for SLIPCOVER): number of mega-examples on which to build the bottom clauses
- **initial\_clauses\_per\_megaex** (values: integer, default value: 1, valid for SLIPCOVER): number of bottom clauses to build for each mega-example (or model or interpretation)
- **d** (values: integer, default value: 1, valid for SLIPCOVER): number of saturation steps when building the bottom clause
- **max\_iter\_structure** (values: integer, default value: 10000, valid for SLIPCOVER): maximum number of theory search iterations
- **background\_clauses** (values: integer, default value: 50, valid for SLIPCOVER): maximum numbers of background clauses
- **maxdepth\_var** (values: integer, default value: 2, valid for SLIPCOVER): maximum depth of variables in clauses (as defined in [5]).
- **neg\_ex** (values: **given**, **cw**, default value: **cw**): if set to **given**, the negative examples in testing are taken from the test folds interpretations, i.e., those examples **ex** stored as **neg(ex)**; if set to **cw**, the negative examples are generated according to the closed world assumption, i.e., all atoms for target predicates that are not positive examples. The set of all atoms is obtained by collecting the set of constants for each type of the arguments of the target predicate.
- **verbosity** (values: integer in [1,3], default value: 1): level of verbosity of the algorithms

## 5.4 Files

The `pack/cplint/prolog/examples` folder in SWI-Prolog home contains some example programs. The subfolder **learning** contains some learning examples. The `pack/cplint/doc` folder in SWI-Prolog home contains this manual in latex, html and pdf.

## 6 License

`cplint` follows the Artistic License 2.0 that you can find in `cplint` root folder. The copyright is by Fabrizio Riguzzi.

The library CUDD for manipulating BDDs has the following license:

Copyright (c) 1995-2004, Regents of the University of Colorado  
All rights reserved.

Redistribution and use in source and binary forms, with or without modification, are permitted provided that the following conditions are met:

- Redistributions of source code must retain the above copyright notice, this list of conditions and the following disclaimer.
- Redistributions in binary form must reproduce the above copyright notice, this list of conditions and the following disclaimer in the documentation and/or other materials provided with the distribution.
- Neither the name of the University of Colorado nor the names of its contributors may be used to endorse or promote products derived from this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

## References

- [1] Elena Bellodi and Fabrizio Riguzzi. EM over binary decision diagrams for probabilistic logic programs. In *Proceedings of the 26th Italian Conference on Computational Logic (CILC2011), Pescara, Italy, 31 August 31-2 September, 2011*, 2011.
- [2] Elena Bellodi and Fabrizio Riguzzi. EM over binary decision diagrams for probabilistic logic programs. Technical Report CS-2011-01, Dipartimento di Ingegneria, Università di Ferrara, Italy, 2011.
- [3] Elena Bellodi and Fabrizio Riguzzi. Expectation Maximization over binary decision diagrams for probabilistic logic programs. *Intel. Data Anal.*, 16(6), 2012.

- [4] Elena Bellodi and Fabrizio Riguzzi. Structure learning of probabilistic logic programs by searching the clause space. *Theory and Practice of Logic Programming*, 2013.
- [5] William W. Cohen. Pac-learning non-recursive prolog clauses. *Artif. Intell.*, 79(1):1–38, 1995.
- [6] L. De Raedt and W. Van Laer. Inductive constraint logic. In *Proceedings of the 6th Conference on Algorithmic Learning Theory (ALT 1995)*, volume 997 of *LNAI*, pages 80–94, Fukuoka, Japan, 1995. Springer.
- [7] Fabrizio Riguzzi and Terrance Swift. Tabling and Answer Subsumption for Reasoning on Logic Programs with Annotated Disjunctions. In *Technical Communications of the International Conference on Logic Programming*, volume 7 of *Leibniz International Proceedings in Informatics (LIPIcs)*, pages 162–171. Schloss Dagstuhl–Leibniz-Zentrum fuer Informatik, 2010.
- [8] J. Vennekens, M. Denecker, and M. Bruynooghe. Representing causal information about a probabilistic process. In *Proceedings of the 10th European Conference on Logics in Artificial Intelligence*, *LNAI*. Springer, September 2006.
- [9] J. Vennekens, Marc Denecker, and Maurice Bruynooghe. CP-logic: A language of causal probabilistic events and its relation to logic programming. *Theory Pract. Log. Program.*, 9(3):245–308, 2009.
- [10] J. Vennekens and S. Verbaeten. Logic programs with annotated disjunctions. Technical Report CW386, K. U. Leuven, 2003.
- [11] J. Vennekens, S. Verbaeten, and M. Bruynooghe. Logic programs with annotated disjunctions. In *International Conference on Logic Programming*, volume 3131 of *LNCS*, pages 195–209. Springer, 2004.